POUČCI O LEPTIRIMA

ZVONKO ČERIN

Sažetak. Ovaj članak opisuje četiri dokaza klasičnog teorema o leptirima za tetive kružnice. Prva tri dokaza koriste standardne tehnike sličnih trokuta, trigonometrije, i potencije točke. Četvrti dokaz je ostvaren upotrebom računala. Istom metodom se dokazuju i tri stupnja poopćenja teorema o leptirima tako da završna faza vrijedi za sve konike (elipse, hiperbole, i parabole).

1. Četiri dokaza Zadatka 122

Na stranici 40 knjige "Izabrani zadaci i teoremi planimetrije" od Škljarskog, Čencova, i Jagloma (vidi [8] iz popisa referenci) nalazi se zadatak 122 koji glasi:

Zadatak 122 (Leptirov teorem). Neka je A, B, C, D, P, i Q šest točaka na kružnici k sa središtem O i neka je S polovište tetive PQ. Neka w označava pravac PQ. Ako je $S = w \cap AB = w \cap CD$, onda je S polovište točaka $U = w \cap AC$ i $V = w \cap BD$ i polovište točaka $X = w \cap AD$ i $Y = w \cap BC$.

Slika 1. Točka S je truplo a trokuti ADS i BCS su krila leptira. Još jedan leptir s istim truplom ima trokute ACS i BDS za krila.

1991 *Mathematics Subject Classification.* Primary 51N20, 51M04, Secondary 14A25, 14Q05.

Key words and phrases. Poučak o lepirima za tetive kružnice, Poučak o lepirima za kružnice i pravce, Poučak o lepirima za konike.

Isti zadatak nalazimo i u knjizi "Trokut i kružnica" od Dominika Palmana (vidi [6]) i to na stranici 197 kao Zadatak 22.15 s imenom "Leptirov teorem".

Prvi dokaz Zadatka 122. Neka je O središte zadane kružnice (vidi Sliku 2). Jer je OS okomito na PQ, za dokaz tvrdnje da je $XS = SY$ dovoljno se uvjeriti da je $\angle XOS = \angle SOY$ (vidi Sliku 3).

Slika 2. Pretpostavka: AB i CD se sijeku u polovištu S od PQ. Tvrdnja: $XS = YS$, gdje su X i Y presjeci AD i BC sa PQ.

Slika 3. Dovoljno je pokazati da je kut $\angle SOX$ jednak kutu $\angle SOY$.

Slika 4. Točka O_1 je polovište tetive AD i projekcija O na AD dok je O_2 polovite tetive BC i projekcija O na BC.

Slika 5. Kutevi $\angle AO_1S$ i $\angle CO_2S$ su jednaki jer su trokuti AO_1S i CO_2S slični.

Neka je O_1 projekcija točke O na tetivu AD i neka je O_2 projekcija točke O na tetivu BC (vidi Sliku 4). Onda su O_1 i O_2 polovišta tih tetiva. Nadalje, vrijedi

$$\angle DAB = \angle DCB$$

i

$$\angle ADC = \angle ABC$$

Slika 6. Oko četverokuta SOO_1X i SOO_2Y mogu se opisati kružnice jer oba imaju dva nasuprotna prava kuta.

Slika 7. Kutevi u O i O_2 trokuta SOY i SO_2Y su isti (nad SY). Kutevi u O i O_1 trokuta SOX i SO_1X su isti (nad SX). Kako je $\angle CO_2S = \angle AO_1S$ slijedi da je $\angle YOS = \angle XOS$ pa je $XS = SY$.

jer su to kutevi nad istim lukom. Zbog toga su trokuti DAS i BCS slični. Ali onda će trokuti O_1AS i O_2CS također biti slični budući da imaju proporcionalne stranice AS i CS odnosno $AO_1 = \frac{AD}{2}$ i $CO_2 = \frac{CB}{2}$ i isti kut (u vrhovima A i C) među njima. Slijedi da je

$$\angle AO_1S = \angle CO_2S$$

(vidi Sliku 5).

Uočimo sada četverokute SOO_1X i SOO_2Y. Oni imaju dva nasuprotna prava kuta pa se oko njih mogu opisati kružnice (vidi Sliku 6).

Iz toga slijedi

$$\angle XO_1S = \angle XOS$$

jer su to kutevi nad istim lukom kružnice (vidi Sliku 7). Slično je

$$\angle YO_2S = \angle YOS.$$

Kako je $\angle CO_2S = \angle AO_1S$ slijedi da je $\angle YOS = \angle XOS$ pa je $XS = SY$.

Drugi dokaz Zadatka 122.

Slika 8. Definicija kuteva α, β, i γ.

Označimo (vidi Slike 8 i 9)

$$\alpha = \angle CSY = \angle XSD,$$

$$\beta = \angle ASX = \angle YSB,$$

$$\gamma = \angle DAS = \angle SCY,$$

$$x = XS \qquad y = SY \qquad z = PS = SQ.$$

Prema teoremu o umnošku odsječaka dviju tetiva kružnice (vidi Sliku 10) vrijedi:

$$DX \cdot XA = PX \cdot XQ = (z - x)(z + x) = z^2 - x^2.$$

Slika 9. Definicije dužina $x = XS$, $y = SY$, i $z = PS = SQ$.

Slika 10. Na PQ i AD primjenim teorem o potenciji (na točku X).

Primjenimo li na trokut DXS teorem sinusa (vidi Sliku 11) dobivamo

$$DX = \frac{x\,\sin\alpha}{\sin[\pi - (\alpha + \beta + \gamma)]} = \frac{x\,\sin\alpha}{\sin(\alpha + \beta + \gamma)}.$$

Slika 11. Na trokut DSX primjenim teorem o sinusima.

Slika 12. Na trokut ASX isto primjenim teorem o sinusima.

Analogno, iz trokuta XAS (vidi Sliku 12), imamo

$$XA = \frac{x \sin\beta}{\sin\gamma}.$$

Zato je

$$DX \cdot XA = \frac{x^2 \sin\alpha \sin\beta}{\sin\gamma \sin(\alpha+\beta+\gamma)} = z^2 - x^2.$$

Odavde možemo izraziti duljinu x:

$$x^2 = \frac{z^2 \sin\gamma \sin(\alpha+\beta+\gamma)}{\sin\alpha \sin\beta + \sin\gamma \sin(\alpha+\beta+\gamma)}.$$

Izraz za duljinu segmenta $SY = y$ dobiva se iz gornjeg izraza za x zamjenom kuteva α i β kutevima β i α. Ali, u tom izrazu kutevi α i β se pojavljuju **simetrično**, pa je

$$y^2 = \frac{z^2 \sin\gamma \sin(\beta+\alpha+\gamma)}{\sin\beta \sin\alpha + \sin\gamma \sin(\beta+\alpha+\gamma)} = x^2.$$

Zaključujemo da je

$$x = y \qquad \text{to jest} \qquad XS = SY.$$

Treći dokaz Zadatka 122.

Slika 13. Treći dokaz koristi projekcije X_1, X_2 i Y_1, Y_2 točaka X i Y i četiri para sličnih pravokutnih trokuta.

Nožišta okomica spuštenih iz točaka X i Y na tetive AB i CD su X_1 i X_2 odnosno Y_2 i Y_1 (vidi Sliku 13).

Uvedimo oznake (vidi Sliku 9):

$$z = PS = SQ, \qquad x = XS, \qquad y = SY.$$

Vrijedi očito:

$$\angle DAB = \angle DCB$$

i

$$\angle ADC = \angle ABC.$$

Osim toga su slijedeći parovi pravokutnih trokuta slični:

$$SXX_1, SYY_2; \qquad SXX_2, SYY_1;$$

$$AXX_1, CYY_1; \qquad DXX_2, BYY_2.$$

Imamo sada:

$$\frac{x}{y} = \frac{XX_1}{YY_2}, \qquad \frac{x}{y} = \frac{XX_2}{YY_1},$$
$$\frac{XX_1}{YY_1} = \frac{AX}{CY}, \qquad \frac{XX_2}{YY_2} = \frac{XD}{YB}.$$

Odatle je dalje:

$$\begin{aligned}\frac{x^2}{y^2} &= \frac{XX_1}{YY_2} \cdot \frac{XX_2}{YY_1}\\ &= \frac{AX \cdot XD}{CY \cdot YB} = \frac{PX \cdot XQ}{PY \cdot YQ}\\ &= \frac{(z-x)(z+x)}{(z+y)(z-y)} = \frac{z^2 - x^2}{z^2 - y^2},\end{aligned}$$

odakle slijedi $x = y$.

Četvrti dokaz Zadatka 122 (pomoću računala). U slijedećem dokazu Zadatka 122 koristimo se računalom i programom Maple V (verzija 5). Ono što moramo natipkati ("input") naznačeno je u redovima koji započinju znakom kursora > dok ostali redovi prikazuju ono što nam izračuna računalo ("output").

Ideja dokaza je da se pretpostavi da je promatrana kružnica jedinična kružnica u pravokutnom koordinatnom sustavu i da se upotrebom analitičke geometrije dokaže tvrdnja.

Bilo koja točka P na jediničnoj kružnici ima koordinate $(\cos\phi, \sin\phi)$ za jedinstveni kut $0 \leq \phi < 2\pi$ (vidi Sliku 14). Iskoristimo li univerzalnu trigonometrijsku substituciju vidimo da su koordinate od P prikazive kao par

$$\left(\frac{1-p^2}{1+p^2}, \frac{2p}{1+p^2}\right),$$

gdje je $p = \tan\frac{\phi}{2}$ realan broj.

SLIKA 14. Točka na jediničnoj kružnici određena je kutem ϕ (f na slici). Apcisa joj je kosinus a ordinata joj je sinus tog kuta.

Zato prvo definiramo funkciju f koja bilo kojem realnom broju a pridružuje uređeni par

$$\left[\frac{1-a^2}{1+a^2}, \frac{2\,a}{1+a^2}\right].$$

```
> f:=proc(a) [(1-a^2)/(1+a^2),2*a/(1+a^2)]:end:
```

Sada mnogo učinkovitije možemo definirati šest točaka leA, leB, leC, leD, leP, i leQ na jediničnoj kružnici (slova le dolaze od "leptir").

```
> leA:=f(a);leB:=f(b);leC:=f(c);leD:=f(d);leP:=f(p);leQ:=f(q);
```

$$leA := [\frac{1-a^2}{1+a^2}, 2\,\frac{a}{1+a^2}]$$

$$leB := [\frac{1-b^2}{1+b^2}, 2\,\frac{b}{1+b^2}]$$

$$leC := [\frac{1-c^2}{1+c^2}, 2\,\frac{c}{1+c^2}]$$

$$leD := [\frac{1-d^2}{1+d^2}, 2\,\frac{d}{1+d^2}]$$

$$leP := [\frac{1-p^2}{1+p^2}, 2\,\frac{p}{1+p^2}]$$

$$leQ := [\frac{1-q^2}{1+q^2},\, 2\,\frac{q}{1+q^2}]$$

Slijedeća funkcija pridružuje paru točaka koordinate polovišta dužine kojoj su te točke krajevi. To polovište ima koordinate koje su aritmetička sredina koordinata krajeva.

```
> ejmid:=proc (A, B)
> local a, u, b, v;
> a:=A[1];u:=A[2];b:=B[1];v:=B[2];
> factor(simplify([(a+b)/2, (u+v)/2])): end:
```

Sada definiramo polovište leS dužine $(leP)(leQ)$.

```
> leS:=ejmid(leP,leQ);
```

$$leS := \left[-\frac{(p\,q-1)\,(1+p\,q)}{(1+p^2)\,(1+q^2)},\, \frac{(q+p)\,(1+p\,q)}{(1+p^2)\,(1+q^2)}\right]$$

U pripremi za što jednostavnije definiranje pravaca potrebnih za određivanje nekih točaka promatrane konfiguracije uvodimo funkciju $ejli$ koja bilo kojem paru različitih točaka pridružuje trojku $[A,\, B,\, C]$ čije koordinate su koeficijenti (linearne!) jednadžbe oblika $A\,x + B\,y + C = 0$ tog pravca.

```
> ejli:=proc (A, B)
> local a, b, c, d;
> a := A[1]; b := A[2]; c := B[1]; d := B[2];
> factor(simplify([b-d, c-a, a*d-b*c])): end:
```

Funkcijom $ejli$ lagano definiramo pravce koji će se pojavljivati u daljnem.

```
> leAB:=ejli(leA,leB);
> leAC:=ejli(leA,leC);leAD:=ejli(leA,leD);
> leBC:=ejli(leC,leB);leBD:=ejli(leD,leB);
> leCD:=ejli(leC,leD);lePQ:=ejli(leP,leQ);
```

$$leAB := \left[-2\,\frac{(b\,a-1)\,(-b+a)}{(1+a^2)\,(1+b^2)},\, 2\,\frac{(-b+a)\,(b+a)}{(1+a^2)\,(1+b^2)},\, -2\,\frac{(b\,a+1)\,(-b+a)}{(1+a^2)\,(1+b^2)}\right]$$

$$leAC := \left[-2\,\frac{(c\,a-1)\,(-c+a)}{(1+a^2)\,(1+c^2)},\, 2\,\frac{(-c+a)\,(c+a)}{(1+a^2)\,(1+c^2)},\, -2\,\frac{(c\,a+1)\,(-c+a)}{(1+a^2)\,(1+c^2)}\right]$$

$$leAD := \left[-2\,\frac{(d\,a-1)\,(-d+a)}{(1+a^2)\,(1+d^2)},\, 2\,\frac{(-d+a)\,(d+a)}{(1+a^2)\,(1+d^2)},\, -2\,\frac{(d\,a+1)\,(-d+a)}{(1+a^2)\,(1+d^2)}\right]$$

$$leBC := \left[2\,\frac{(c\,b-1)\,(-c+b)}{(1+c^2)\,(1+b^2)},\, -2\,\frac{(-c+b)\,(b+c)}{(1+c^2)\,(1+b^2)},\, 2\,\frac{(c\,b+1)\,(-c+b)}{(1+c^2)\,(1+b^2)}\right]$$

$$leBD := \left[2\,\frac{(d\,b-1)\,(-d+b)}{(1+d^2)\,(1+b^2)},\, -2\,\frac{(-d+b)\,(b+d)}{(1+d^2)\,(1+b^2)},\, 2\,\frac{(d\,b+1)\,(-d+b)}{(1+d^2)\,(1+b^2)}\right]$$

$$leCD := \left[-2\,\frac{(d\,c-1)\,(-d+c)}{(1+c^2)\,(1+d^2)},\, 2\,\frac{(-d+c)\,(d+c)}{(1+c^2)\,(1+d^2)},\, -2\,\frac{(d\,c+1)\,(-d+c)}{(1+c^2)\,(1+d^2)}\right]$$

$$lePQ := \left[-2\,\frac{(p\,q-1)\,(-q+p)}{(1+p^2)\,(1+q^2)},\ 2\,\frac{(-q+p)\,(q+p)}{(1+p^2)\,(1+q^2)},\ -2\,\frac{(1+p\,q)\,(-q+p)}{(1+p^2)\,(1+q^2)}\right]$$

Naravno, treba nam i funkcija $ejin$ koja paru pravaca pridružuje točku njihovog presjeka ili nas izvještava da su ti pravci paralelni.

```
> ejin:=proc (m1, m2)
> local a, b, c, u, v, w, iz1, iz2, iz3;
> a := m1[1]; b := m1[2]; c := m1[3];
> u := m2[1]; v := m2[2]; w := m2[3];
> iz1 := a*v-b*u; iz2 := b*w-c*v; iz3 :=c*u-a*w;
> if iz1 <> 0 then
> factor(simplify([iz2/iz1, iz3/iz1])):
> else print(‘pravci su paralelni‘) fi end:
```

Slijede definicije točaka leX, leY, i $leS0$ kao presjeka slijedećih parova pravaca: ($lePQ$, $leAD$), ($lePQ$, $leBC$), i ($leAB$, $leCD$).

```
> leX:=ejin(lePQ,leAD);leY:=ejin(lePQ,leBC);leS0:=ejin(leAB,leCD);
```

$$leX := \left[-\frac{q\,d\,a+q+p\,d\,a+p-d-a-p\,q\,d-p\,q\,a}{-p\,q\,d-p\,q\,a+d+a+q\,d\,a-q+p\,d\,a-p},\right.$$
$$\left. 2\,\frac{d\,a-p\,q}{-p\,q\,d-p\,q\,a+d+a+q\,d\,a-q+p\,d\,a-p}\right]$$

$$leY := \left[-\frac{q\,c\,b+q+p\,c\,b+p-b-c-p\,q\,b-p\,q\,c}{-p\,q\,b-p\,q\,c+b+c+q\,c\,b-q+p\,c\,b-p},\right.$$
$$\left. 2\,\frac{c\,b-p\,q}{-p\,q\,b-p\,q\,c+b+c+q\,c\,b-q+p\,c\,b-p}\right]$$

$$leS0 := \left[-\frac{-b\,d\,c-b-a\,d\,c-a+b\,a\,d+b\,a\,c+d+c}{b\,a\,d+b\,a\,c-d-c-b\,d\,c+b-a\,d\,c+a},\right.$$
$$\left. 2\,\frac{b\,a-d\,c}{b\,a\,d+b\,a\,c-d-c-b\,d\,c+b-a\,d\,c+a}\right]$$

Pretpostavka da se pravci $leAB$ i $leCD$ sijeku u točki leS (tj. da se točke leS i $leS0$ podudaraju) vodi na zaključak da postoje jednoznačno određene vrijednosti parametara a i c kada se to postiže.

```
> rj:=solve({leS0[1]=leS[1],leS0[2]=leS[2]},{a,c});
```

$$rj := \{a = \frac{p^2\,q\,b-2\,p^2\,q^2-p^2+p\,q^2\,b+p\,b-q^2+q\,b}{b\,p^2-p^2\,q-p-p\,q^2-q+b\,q^2+2\,b},$$
$$c = \frac{-p^2-2\,p^2\,q^2-q^2+q\,d+p\,q^2\,d+p^2\,q\,d+p\,d}{2\,d-p\,q^2-q-p-p^2\,q+d\,q^2+d\,p^2}\}$$

Na kraju, treba nam i funkcija $ejkdis$ koja paru točaka pridružuje kvadrat njihove udaljenosti u standardnoj Euklidskoj metrici na ravnini.

```
>  ejkdis:=proc (A, B)
>  local a, u, b, v;
>  a := A[1]; u := A[2]; b :=B[1]; v := B[2];
>  factor(simplify((a-b)^2 + (u-v)^2)):  end:
```

Odredimo razliku kvadrata udaljenosti parova točaka (leX, leS) i (leY, leS).

```
>  con:=factor(simplify(ejkdis(leS,leX)-ejkdis(leS,leY)));
```

$$con := -4((p\,q\,c - q\,c\,b - p^2\,q - p\,q^2 + c\,q^2 + c\,p^2 + d\,q^2 + d\,p^2 + a\,q^2 + a\,p^2 + p\,q\,a - q\,d\,a - p\,d\,b + b\,q^2 + p\,q\,d + b\,d\,c - 2\,p^3\,q^2 - c\,q\,d + 2\,c\,p^2\,q^2 + 2\,d\,p^2\,q^2 + 2\,a\,p^2\,q^2 + b\,p^2 + 2\,b\,p^2\,q^2 + p\,q\,b - p\,c\,b - p\,d\,a - a\,p\,b - p^3 - p\,d\,c - q^3 - 2\,p^2\,q^3 + b\,a\,d + a\,d\,c + b\,a\,c - 2\,a\,q\,p\,c\,b - a\,q\,p^3\,d\,c + a\,d\,p^3\,c\,b + a\,d\,q^3\,c\,b - a\,p\,q^3\,c\,b - a\,p^2\,q^2\,c\,b - a\,p^3\,q\,c\,b + a\,p^2\,q\,b\,d\,c - q\,b\,d + a\,p\,q^2\,b\,d\,c - a\,d\,c\,p^2\,q^2 + 2\,a\,q\,b\,d\,c + b\,p^2\,q^3\,d + b\,p^3\,q^2\,d - a\,c\,p\,q^3\,d + 2\,a\,p\,b\,d\,c - 2\,a\,q\,d\,p\,c - a\,q\,c - a\,p\,c + a\,p^2\,q^3\,c + a\,p^3\,q^2\,c - p^3\,q^3\,d - p^3\,q^3\,a + d\,p^3\,q^2\,c + d\,p^2\,q^3\,c + p^3\,q^2\,a\,d + q^3\,d\,a\,p^2 - b\,p\,q^3\,d\,c - b\,p^3\,q\,d\,c + p^2\,q^3\,c\,b + p^3\,q^2\,c\,b - p^3\,q^3\,b - p^3\,q^3\,c - 2\,p\,c\,b\,q\,d - a\,b\,p\,q^3\,d - a\,b\,p^3\,q\,d - 2\,a\,q\,d\,p\,b - a\,b\,d\,p^2\,q^2 + p^3\,a\,q^2\,b + p^2\,q^3\,a\,b - b\,q\,a - b\,d\,p^2\,q^2\,c)(-b\,a\,d + b\,a\,c - a\,d\,c + p\,d\,a + q\,d\,a - p\,q\,a + p\,q\,b - p\,q\,d - q\,c\,b + b\,d\,c + p\,q\,c - p\,c\,b))/((-p\,q\,b - p\,q\,c + b + c + q\,c\,b - q + p\,c\,b - p)^2 (-p\,q\,d - p\,q\,a + d + a + q\,d\,a - q + p\,d\,a - p)^2)$$

Tako smo dokaz Zadatka 122 sveli na provjeru da li je izraz con jednak nuli kada u njega uvrstimo vrijednosti za a i c iz rj.

```
>  factor(simplify(subs(rj,con)));
```

$$0$$

Kako je doista dobivena nula naš dokaz pomoću računala je uspješno završen.

2. Malo poopćenje Zadatka 122

Malo poopćeni Zadatak 122. Neka je A, B, C, D, P, i Q šest točaka na kružnici k sa središtem O i neka je S polovište tetive PQ. Neka w označava pravac PQ. Ako je S polovište točaka $H = w \cap AB$ i $K = w \cap CD$, onda je S također polovište točaka $U = w \cap AC$ i $V = w \cap BD$ i polovište točaka $X = w \cap AD$ i $Y = w \cap BC$.

Dokaz malo poopćenog Zadatka 122.

I dalje ćemo koristiti računalo i izmjeniti četvrti dokaz Zadatka 122 tako da dobijemo zaključak gornjeg poopćenja (vidi Sliku 15).

Prvo nađemo koordinate točaka iz iskaza.

```
>  leU:=ejin(lePQ,leAC);leV:=ejin(lePQ,leBD);
>  leH:=ejin(lePQ,leAB);leK:=ejin(lePQ,leCD);
```

Slika 15. Slika za Malo poopćeni Zadatak 122.

$$leU := [-\frac{a\,q\,c+q+a\,p\,c+p-c-a-p\,q\,c-p\,q\,a}{-p\,q\,c-p\,q\,a+c+a+a\,q\,c-q+a\,p\,c-p},$$
$$2\,\frac{c\,a-p\,q}{-p\,q\,c-p\,q\,a+c+a+a\,q\,c-q+a\,p\,c-p}]$$
$$leV := [-\frac{d\,q\,b+q+d\,p\,b+p-d-b-p\,q\,d-p\,q\,b}{-p\,q\,d-p\,q\,b+d+b+d\,q\,b-q+d\,p\,b-p},$$
$$2\,\frac{d\,b-p\,q}{-p\,q\,d-p\,q\,b+d+b+d\,q\,b-q+d\,p\,b-p}]$$
$$leH := [-\frac{q\,b\,a+q+p\,b\,a+p-b-a-p\,q\,b-p\,q\,a}{-p\,q\,b-p\,q\,a+b+a+q\,b\,a-q+p\,b\,a-p},$$
$$2\,\frac{b\,a-p\,q}{-p\,q\,b-p\,q\,a+b+a+q\,b\,a-q+p\,b\,a-p}]$$
$$leK := [-\frac{d\,q\,c+q+d\,p\,c+p-d-c-p\,q\,d-p\,q\,c}{-p\,q\,d-p\,q\,c+d+c+d\,q\,c-q+d\,p\,c-p},$$
$$2\,\frac{d\,c-p\,q}{-p\,q\,d-p\,q\,c+d+c+d\,q\,c-q+d\,p\,c-p}]$$

Ovaj puta vidimo da se uvjet da je točka S polovište dužine HK može riješiti po parametru d.

```
> rjm:=solve({ejmid(leH,leK)[1]=leS[1],
> ejmid(leH,leK)[2]=leS[2]},{d});
```

$$
\begin{aligned}
&rjm := \{d = (p\,c\,b + p^2\,q - b\,q^2 + q\,c\,b - c\,q^2 - p\,q\,c - a\,p^2 - b\,p^2 + p\,q^2 - \\
&2\,b\,p^2\,q^2 - c\,p^2 - 2\,c\,p^2\,q^2 - a\,q^2 - 2\,a\,p^2\,q^2 - p\,q\,a - b\,a\,c - p\,q\,b + q^3 + p^3\,q^3\,c \\
&+ p^3\,q^3\,b + 2\,p^2\,q^3 + 2\,p^3\,q^2 - p^2\,q^3\,c\,b - p^3\,q^2\,c\,b + p^3 - p^2\,a\,q^3\,c - p^3\,a\,q^2\,b - \\
&p^3\,a\,q^2\,c - p^2\,a\,q^3\,b + p^3\,q^3\,a + q\,b\,a + p\,b\,a + a\,p\,q^3\,c\,b + a\,p^2\,q^2\,c\,b + a\,p^3\,q\,c\,b \\
&+ 2\,a\,q\,p\,c\,b + a\,p\,c + a\,q\,c)/(-p\,a - p\,b - p\,c - q\,a - q\,b - q\,c - 2\,p\,q\,c\,b + p^2 \\
&+ q^2 + p\,q + 2\,p^2\,q^2 + p\,q^2\,b\,a\,c + p^2\,q\,b\,a\,c - b\,a\,p^2\,q^2 + 2\,q\,b\,a\,c + 2\,p\,b\,a\,c \\
&+ b\,a + c\,a - 2\,p\,q\,c\,a + c\,b + p^3\,b\,a\,c - p^3\,q^3 - p^3\,q\,c\,b - p\,q^3\,c\,b - a\,p\,q^3\,c \\
&- p^2\,q^2\,b\,c - a\,p^3\,q\,c + a\,q^3\,c\,b - p^2\,q^2\,a\,c - 2\,p\,q\,b\,a + p^3\,q^2\,b + p^3\,q^2\,a \\
&- p^3\,q\,b\,a + p^2\,q^3\,b + p^2\,q^3\,a - p\,q^3\,b\,a + p^3\,q^2\,c + p^2\,q^3\,c)\}
\end{aligned}
$$

Sada samo provjerimo da li su razlike kvadrata udaljenosti od X do S i od Y do S odnosno od U do S i od V do S jednake nuli.

```
> factor(simplify(subs(rjm,ejkdis(leX,leS)-ejkdis(leS,leY))));
> factor(simplify(subs(rjm,ejkdis(leU,leS)-ejkdis(leS,leV))));
```

$$0$$
$$0$$

3. Leptirov teorem za kružnice i pravce

Prethodno smo u iskazu Zadatka 122 i u njegovom malom poopćemju uvijek pretpostavljali da je PQ tetiva zadane kružnice. Sada ćemo tu tetivu zamijeniti bilo kakvim pravcem a točka S će biti projekcija središta kružnice na taj pravac (vidi Sliku 16 i [9]).

Leptirov teorem za kružnice i pravce. Neka je A, B, C, i D četiri točaka na kružnici k sa središtem O i neka je S projekcija od O na pravac w u ravnini kružnice. Ako je S polovište točaka $H = w \cap AB$ i $K = w \cap CD$, onda je S također polovište točaka $U = w \cap AC$ i $V = w \cap BD$ i polovište točaka $X = w \cap AD$ i $Y = w \cap BC$.

Dokaz Leptirovog teorema za kružnice i pravce. Ovaj puta prvo zadamo pravac trojkom realnih brojeva.

```
> lew:=[p,q,r];
```

$$lew := [p,\, q,\, r]$$

Sada nam treba funkcija koja točki i pravcu pridružuje projekciju te točke na zadani pravac.

Slika 16. Leptirov teorem za kružnice i pravce.

```
> ejproj:=proc (A, m)
> local a, b, c, u, v;
> u := A[1]; v := A[2];
> a := m[1]; b := m[2]; c := m[3];
> factor(simplify(([-(b*a*v-u*b^2+a*c)/(a^2+b^2),
> (-b*c+a^2*v-a*u*b)/(a^2+b^2)]))): end:
```

Slijedi zadavanje točke S.

```
> leS:=ejproj([0,0],lew);
```

$$leS := [-\frac{p\,r}{p^2+q^2},\ -\frac{q\,r}{p^2+q^2}]$$

Koordinate točaka X, Y, U, V, H, i K su:

```
> leX:=ejin(lew,leAD);leY:=ejin(lew,leBC);
> leU:=ejin(lew,leAC);leV:=ejin(lew,leBD);
> leH:=ejin(lew,leAB);leK:=ejin(lew,leCD);
```

$$leX := [-\frac{q\,d\,a+q+r\,d+r\,a}{p\,d+p\,a+q\,d\,a-q},\ \frac{-r\,d\,a+r+p\,d\,a+p}{p\,d+p\,a+q\,d\,a-q}]$$

$$leY := [-\frac{q\,c\,b+q+r\,c+r\,b}{p\,c+p\,b+q\,c\,b-q},\ \frac{-r\,c\,b+r+p\,c\,b+p}{p\,c+p\,b+q\,c\,b-q}]$$

$$leU := [-\frac{a\,q\,c+q+r\,c+r\,a}{p\,c+p\,a+a\,q\,c-q},\ \frac{-r\,c\,a+r+a\,p\,c+p}{p\,c+p\,a+a\,q\,c-q}]$$

$$leV := [-\frac{d\,q\,b + q + r\,d + r\,b}{p\,d + p\,b + d\,q\,b - q}, \frac{-r\,d\,b + r + d\,p\,b + p}{p\,d + p\,b + d\,q\,b - q}]$$

$$leH := [-\frac{q\,b\,a + q + r\,b + r\,a}{p\,b + p\,a + q\,b\,a - q}, \frac{-r\,b\,a + r + p\,b\,a + p}{p\,b + p\,a + q\,b\,a - q}]$$

$$leK := [-\frac{d\,q\,c + q + r\,d + r\,c}{p\,d + p\,c + d\,q\,c - q}, \frac{-r\,d\,c + r + d\,p\,c + p}{p\,d + p\,c + d\,q\,c - q}]$$

Opet se uvjet da je S polovište segmenta HK može riješiti po parametru d:

```
> rjm:=solve({ejmid(leH,leK)[1]=leS[1],
> ejmid(leH,leK)[2]=leS[2]},{d});
```

$$rjm := \{d = -(-2\,p^2\,q + p\,q^2\,a + p\,q^2\,b + p\,q^2\,c + p\,q^2\,b\,a\,c + p^3\,a + p^3\,c + p^3\,b + p^3\,b\,a\,c - 2\,q^3 + 2\,r\,q\,b\,a\,p - r\,b\,a\,p^2\,c + r\,b\,a\,c\,q^2 + 2\,r\,q\,p\,b\,c + 2\,r\,q\,p\,a\,c - r\,b\,q^2 - r\,a\,q^2 - r\,c\,q^2 + r\,p^2\,c + r\,b\,p^2 + r\,a\,p^2 - 2\,r\,q\,p)/(p\,q^2 + p\,q^2\,b\,a + p\,q^2\,b\,c + p\,q^2\,a\,c + p^3\,c\,b - 2\,r\,b\,a\,q\,c\,p + 2\,r\,q\,p\,c + r\,a\,c\,q^2 + r\,b\,c\,q^2 - r\,a\,p^2\,c - r\,b\,p^2\,c - r\,b\,a\,p^2 + r\,b\,a\,q^2 + 2\,r\,q\,b\,p + 2\,r\,q\,a\,p - r\,q^2 + 2\,a\,q^3\,c\,b + 2\,p^2\,q\,b\,a\,c + r\,p^2 + p^3 + p^3\,c\,a + p^3\,b\,a)\}$$

Završetak dokaza je isti.

```
> factor(simplify(subs(rjm,ejkdis(leX,leS)-ejkdis(leS,leY))));
> factor(simplify(subs(rjm,ejkdis(leU,leS)-ejkdis(leS,leV))));
```

$$0$$
$$0$$

4. Leptirov teorem za konike

Sada ćemo opisati proširenje Leptirovog teorema za kružnice i pravce na konike (ili čunjosječnice) (vidi [1]). Nažalost ne mogu se koristiti svi pravci već se moramo ograničiti samo na pravce koji su okomiti na pravac simetrije zadane konike.

Leptirov teorem za konike. Neka je S točka na pravcu simetrije z za koniku k i neka je w pravac kroz S okomit na z. Neka su A, B, C, i D različite točke na k i neka nam H, K, U, V, X, and Y označavaju presjeke pravca w s pravcima AB, CD, AC, BD, AD, and BC, redom. Ako je točka S polovište jednog od segmenata HK, UV, i XY, onda je ona polovište svih njih.

Primjetimo da Leptirov teorem za konike povlači Leptirov teorem za kružnice i pravce jer je svaki pravac koji prolazi središtem kružnice njen pravac simetrije.

Dokaz leptirovog teorema za konike. Prisjetimo se da ako uzmemo fokus konike k za pol (ishodište) i glavnu os konike (tj. pravac simetrije kroz fokus) m za polarnu os polarnog koordinatnog sustava onda zadana konika k ima jednadžbu

$$\varrho = \frac{p}{1 + \varepsilon\,\cos\vartheta},$$

gdje je ϱ polarni radijus, ϑ je polarni kut, a p i ε su nenegativni realni brojevi. (Na Slici 17 je prikazan slučaj parabole kada je $\varepsilon = 1$).

Slika 17. Prikaz parabole u polarnom koordinatnom sustavu.

Dakle, u pridruženom pravokutnom koordinatnom sustavu točke A, B, C, i D imaju koordinate
$$\left(\frac{p\,\cos\vartheta}{1+\varepsilon\,\cos\vartheta}, \frac{p\,\sin\vartheta}{1+\varepsilon\,\cos\vartheta}\right),$$
gdje je polarni kut ϑ jednak redom kutevima α, β, γ, i δ. Mogli bi smo nastaviti upotrebljavati trigonometrijske funkcije ali je mnogo jednostavnije opet iskoristiti univerzalnu trigonometrijsku supstituciju i pisati
$$\cos\alpha = \frac{1-a^2}{1+a^2}, \qquad \sin\alpha = \frac{2\,a}{1+a^2},$$
i slično za preostale tri točke. Zaključujemo da točke A, B, C, i D imaju koordinate
$$\left(\frac{p\,(1-t^2)}{\varepsilon\,(1-t^2)+t^2+1}, \frac{2\,p\,t}{\varepsilon\,(1-t^2)+t^2+1}\right)$$
za parametar t jednak a, b, c, i d.

Zato sada koristimo slijedeću funkciju:

```
> f:=proc(t)
> [p*(1-t^2)/(epsilon*(1-t^2)+t^2+1),
> 2*p*t/(epsilon*(1-t^2)+t^2+1)]:end:
```

Točke A, B, C, i D zadane su sa:

```
> leA:=f(a);leB:=f(b);leC:=f(c);leD:=f(d);
```

$$leA := [\frac{p\,(1-a^2)}{\varepsilon\,(1-a^2)+a^2+1},\, 2\,\frac{p\,a}{\varepsilon\,(1-a^2)+a^2+1}]$$

$$leB := [\frac{p\,(1-b^2)}{\varepsilon\,(1-b^2)+b^2+1},\, 2\,\frac{p\,b}{\varepsilon\,(1-b^2)+b^2+1}]$$

$$leC := [\frac{p\,(1-c^2)}{\varepsilon\,(1-c^2)+c^2+1},\, 2\,\frac{p\,c}{\varepsilon\,(1-c^2)+c^2+1}]$$

$$leD := [\frac{p\,(1-d^2)}{\varepsilon\,(1-d^2)+d^2+1},\, 2\,\frac{p\,d}{\varepsilon\,(1-d^2)+d^2+1}]$$

Dokažimo prvo gornji teorem u slučaju kada je pravac z glavna os m konike k (vidi Sliku 18). Točka S onda ima koordinate $(s, 0)$ za neki realan broj s a pravac w ima jednadžbu $x = s$.

Slika 18. Leptirov teorem za elipsu kada je pravac w okomit na glavnu os simetrije z.

```
> leS:=[s,0];
```

$$leS := [s,\, 0]$$

```
> lew:=[1,0,-s];
```

$$lew := [1,\, 0,\, -s]$$

Slijedeći korak je određivanje šest pravaca koji spajaju točke A, B, C, i D.

```
> leAB:=ejli(leA,leB);leAC:=ejli(leA,leC);leAD:=ejli(leA,leD);
> leBC:=ejli(leC,leB);leBD:=ejli(leD,leB);leCD:=ejli(leC,leD);
```

$$leAB := [2\,\frac{p\,(a-b)\,(a\,b\,\varepsilon - b\,a + \varepsilon + 1)}{(-\varepsilon + \varepsilon\,a^2 - a^2 - 1)\,(-\varepsilon + \varepsilon\,b^2 - b^2 - 1)},$$
$$2\,\frac{p\,(a-b)\,(a+b)}{(-\varepsilon + \varepsilon\,a^2 - a^2 - 1)\,(-\varepsilon + \varepsilon\,b^2 - b^2 - 1)},$$
$$-2\,\frac{p^2\,(b\,a+1)\,(a-b)}{(-\varepsilon + \varepsilon\,a^2 - a^2 - 1)\,(-\varepsilon + \varepsilon\,b^2 - b^2 - 1)}]$$

$$leAC := [2\,\frac{p\,(a-c)\,(a\,c\,\varepsilon - c\,a + \varepsilon + 1)}{(-\varepsilon + \varepsilon\,a^2 - a^2 - 1)\,(-\varepsilon + \varepsilon\,c^2 - c^2 - 1)},$$
$$2\,\frac{p\,(a-c)\,(a+c)}{(-\varepsilon + \varepsilon\,a^2 - a^2 - 1)\,(-\varepsilon + \varepsilon\,c^2 - c^2 - 1)},$$
$$-2\,\frac{p^2\,(c\,a+1)\,(a-c)}{(-\varepsilon + \varepsilon\,a^2 - a^2 - 1)\,(-\varepsilon + \varepsilon\,c^2 - c^2 - 1)}]$$

$$leAD := [2\,\frac{p\,(a-d)\,(a\,d\,\varepsilon - d\,a + \varepsilon + 1)}{(-\varepsilon + \varepsilon\,a^2 - a^2 - 1)\,(-\varepsilon + \varepsilon\,d^2 - d^2 - 1)},$$
$$2\,\frac{p\,(a-d)\,(a+d)}{(-\varepsilon + \varepsilon\,a^2 - a^2 - 1)\,(-\varepsilon + \varepsilon\,d^2 - d^2 - 1)},$$
$$-2\,\frac{p^2\,(d\,a+1)\,(a-d)}{(-\varepsilon + \varepsilon\,a^2 - a^2 - 1)\,(-\varepsilon + \varepsilon\,d^2 - d^2 - 1)}]$$

$$leBC := [-2\,\frac{p\,(b-c)\,(b\,c\,\varepsilon - c\,b + \varepsilon + 1)}{(-\varepsilon + \varepsilon\,c^2 - c^2 - 1)\,(-\varepsilon + \varepsilon\,b^2 - b^2 - 1)},$$
$$-2\,\frac{p\,(b-c)\,(b+c)}{(-\varepsilon + \varepsilon\,c^2 - c^2 - 1)\,(-\varepsilon + \varepsilon\,b^2 - b^2 - 1)},$$
$$2\,\frac{p^2\,(c\,b+1)\,(b-c)}{(-\varepsilon + \varepsilon\,c^2 - c^2 - 1)\,(-\varepsilon + \varepsilon\,b^2 - b^2 - 1)}]$$

$$leBD := [-2\,\frac{p\,(b-d)\,(b\,d\,\varepsilon - d\,b + \varepsilon + 1)}{(-\varepsilon + \varepsilon\,d^2 - d^2 - 1)\,(-\varepsilon + \varepsilon\,b^2 - b^2 - 1)},$$
$$-2\,\frac{p\,(b-d)\,(b+d)}{(-\varepsilon + \varepsilon\,d^2 - d^2 - 1)\,(-\varepsilon + \varepsilon\,b^2 - b^2 - 1)},$$
$$2\,\frac{p^2\,(d\,b+1)\,(b-d)}{(-\varepsilon + \varepsilon\,d^2 - d^2 - 1)\,(-\varepsilon + \varepsilon\,b^2 - b^2 - 1)}]$$

$$leCD := [2\,\frac{p\,(c-d)\,(c\,d\,\varepsilon - d\,c + \varepsilon + 1)}{(-\varepsilon + \varepsilon\,c^2 - c^2 - 1)\,(-\varepsilon + \varepsilon\,d^2 - d^2 - 1)},$$
$$2\,\frac{p\,(c-d)\,(c+d)}{(-\varepsilon + \varepsilon\,c^2 - c^2 - 1)\,(-\varepsilon + \varepsilon\,d^2 - d^2 - 1)},$$
$$-2\,\frac{p^2\,(d\,c+1)\,(c-d)}{(-\varepsilon + \varepsilon\,c^2 - c^2 - 1)\,(-\varepsilon + \varepsilon\,d^2 - d^2 - 1)}]$$

Sada su na redu točke H, K, U, V, X, i Y:

```
> leX:=ejin(lew,leAD);leY:=ejin(lew,leBC);
> leU:=ejin(lew,leAC);leV:=ejin(lew,leBD);
> leH:=ejin(lew,leAB);leK:=ejin(lew,leCD);
```

$$leX := [s,\, -\frac{s\,a\,d\,\varepsilon - s\,d\,a + s\,\varepsilon + s - p\,d\,a - p}{a+d}]$$
$$leY := [s,\, -\frac{s\,b\,c\,\varepsilon - s\,c\,b + s\,\varepsilon + s - p\,c\,b - p}{b+c}]$$
$$leU := [s,\, -\frac{s\,a\,c\,\varepsilon - s\,c\,a + s\,\varepsilon + s - p\,c\,a - p}{a+c}]$$
$$leV := [s,\, -\frac{s\,b\,d\,\varepsilon - s\,d\,b + s\,\varepsilon + s - p\,d\,b - p}{b+d}]$$
$$leH := [s,\, -\frac{s\,a\,b\,\varepsilon - s\,b\,a + s\,\varepsilon + s - p\,b\,a - p}{a+b}]$$
$$leK := [s,\, -\frac{s\,c\,d\,\varepsilon - s\,d\,c + s\,\varepsilon + s - p\,d\,c - p}{c+d}]$$

Uvjet da je točka S polovište od HK opet se može riješiti po parametru d:

```
> rj:=solve({ejmid(leH,leK)[1]=leS[1],
> ejmid(leH,leK)[2]=leS[2]},{d});
```

$$rj := \{d = -$$
$$\frac{s\,a\,b\,\varepsilon\,c - s\,b\,a\,c + s\,c\,\varepsilon + s\,c - p\,b\,a\,c - p\,c + s\,a\,\varepsilon + s\,b\,\varepsilon + s\,a + s\,b - p\,a - p\,b}{s\,a\,b\,\varepsilon - s\,b\,a + s\,\varepsilon + s - p\,b\,a - p + s\,a\,c\,\varepsilon + s\,b\,c\,\varepsilon - s\,c\,a - s\,c\,b - p\,c\,a - p\,c\,b}\}$$

Zadnji korak nam je već poznat:

```
> factor(simplify(subs(rj,ejkdis(leX,leS)-ejkdis(leS,leY))));
> factor(simplify(subs(rj,ejkdis(leU,leS)-ejkdis(leS,leV))));
```

$$0$$
$$0$$

Ako je k parabola, onda je dokaz završen budući da je jedini pravac simetrije parabole njezina glavna os (vidi Sliku 19).

Kada je k bilo elipsa ili hiperbola, moramo također promatrati i drugi pravac simetrije (okomicu n na m u centru O od k) kao drugu (i stoga posljednju, ako k nije kružnica) mogućnost za pravac z. Slučaj kada je k kružnica smo već dokazali

Slika 19. Leptirov teorem za parabolu kada je pravac w okomit na glavnu os simetrije m koja je njena jedina os simetrije.

u Leptirovom teoremu za kružnice i pravce. Dakle, kada je $z = n$, onda točka S ima koordinate

$$(\frac{p\,\varepsilon}{\varepsilon^2 - 1}, s)$$

za neki realan broj s a pravac w ima jednadžbu $y = s$ (vidi Sliku 20).

```
> leS:=[p*epsilon/(epsilon^2-1),s];
```

$$leS := [\frac{p\,\varepsilon}{\varepsilon^2 - 1},\ s]$$

```
> lew:=[0,1,-s];
```

$$lew := [0,\ 1,\ -s]$$

```
> leX:=ejin(lew,leAD);leY:=ejin(lew,leBC);
> leU:=ejin(lew,leAC);leV:=ejin(lew,leBD);
> leH:=ejin(lew,leAB);leK:=ejin(lew,leCD);
```

$$leX := [-\frac{-p\,d\,a - p + s\,a + s\,d}{a\,d\,\varepsilon - d\,a + \varepsilon + 1},\ s]$$

$$leY := [\frac{p\,c\,b + p - s\,b - s\,c}{b\,c\,\varepsilon - c\,b + \varepsilon + 1},\ s]$$

Slika 20. Leptirov teorem za hiperbolu kada je pravac w paralelan glavnoj osi simetrije m.

$$leU := [\frac{p\,c\,a + p - s\,a - s\,c}{a\,c\,\varepsilon - c\,a + \varepsilon + 1},\ s]$$

$$leV := [-\frac{-p\,d\,b - p + s\,b + s\,d}{b\,d\,\varepsilon - d\,b + \varepsilon + 1},\ s]$$

$$leH := [\frac{p\,b\,a + p - s\,a - s\,b}{a\,b\,\varepsilon - b\,a + \varepsilon + 1},\ s]$$

$$leK := [-\frac{-p\,d\,c - p + s\,c + s\,d}{c\,d\,\varepsilon - d\,c + \varepsilon + 1},\ s]$$

```
> rj:=solve({ejmid(leH,leK)[1]=leS[1],
> ejmid(leH,leK)[2]=leS[2]},{d});
```

$$rj := \{d = (-s\,a\,\varepsilon + 2\,p + s\,b\,a\,c - s\,a - s\,a\,b\,\varepsilon\,c - s\,c - s\,b + s\,c\,\varepsilon^2 + s\,b\,\varepsilon^2 - s\,c\,\varepsilon - s\,b\,\varepsilon + 4\,\varepsilon\,p + 2\,p\,\varepsilon^2 - s\,b\,a\,c\,\varepsilon^2 + s\,a\,b\,\varepsilon^3\,c + s\,a\,\varepsilon^3 + s\,a\,\varepsilon^2 + s\,c\,\varepsilon^3 + s\,b\,\varepsilon^3) \Big/ (s\,a\,b\,\varepsilon - s\,b\,a - s\,c\,a - s\,c\,b + s\,b\,c\,\varepsilon + s\,\varepsilon^2\,c\,b + s + s\,a\,c\,\varepsilon - s\,a\,b\,\varepsilon^3 + s\,\varepsilon - 4\,p\,b\,a\,c\,\varepsilon + 2\,p\,b\,a\,c - s\,\varepsilon^2 + s\,b\,a\,\varepsilon^2 + s\,a\,c\,\varepsilon^2 - s\,b\,\varepsilon^3\,c - s\,a\,\varepsilon^3\,c + 2\,p\,b\,a\,c\,\varepsilon^2 - s\,\varepsilon^3)\}$$

```
> factor(simplify(subs(rj,ejkdis(leX,leS)-ejkdis(leS,leY))));
> factor(simplify(subs(rj,ejkdis(leU,leS)-ejkdis(leS,leV))));
```

$$0$$

$$0$$

Reference

[1] Zvonko Čerin, *A generalization of the butterfly theorem for circles to conics*, Mathematical Communications (Osijek), (prijavljeno).

[2] H. S. M. Coxeter, *Projective geometry*, Blaisdell, New York, 1964., p. 78.

[3] H. Eves, *A survey of geometry*, Allyn and Bacon, Boston, 1963., p. 171.

[4] Larry Hoehn, *A New Proof of the Double Butterfly Theorem*, Math. Mag., **63** (1990), 256–257.

[5] Murray S. Klamkin, *An extension of the butterfly problem*, Math. Mag., **38** (1965), 206–208.

[6] Dominik Palman, *Trokut i kružnica*, Element, Zagreb, 1994.

[7] J. Sledge, *A generalization of the butterfly theorem*, J. of Undergraduate Math., **5** (1973), 3–4.

[8] D. O. Škljarski, N. N. Čencov, I. M. Jaglom, *Izabrani zadatci i teoremi planimetrije*, Nauka, Moskva, 1967. (na ruskom).

[9] Vladimir Volenec, *A generalization of the butterfly theorem*, Mathematical Communications (Osijek), **6** (2001), 157–160.

Kopernikova 7, 10010 Zagreb, CROATIA, Europe
E-mail address: cerin@math.hr